Some formulas for Chapter 13

Let r (t) be the position (or a parameterization of a curve)

In[31]:=
```
Clear[t, v, a, UnitTangent, UnitNormal, κ, speed]
```

In[32]:=
```
r = {t, t^2 , t^3}
```

Out[32]= $\{t, t^2, t^3\}$

The velocity is dr / dt, and the acceleration is $d^2 r / dt^2 = dv / dt$

In[33]:=
```
v = D[r, t]
```

Out[33]= $\{1, 2t, 3t^2\}$

In[34]:=
```
a = D[v, t]
```

Out[34]= $\{0, 2, 6t\}$

Note that there are simplifications in the above, mostly from sin^2 + cos^2 = 1

The speed is the length of the velocity,
and the (arc) length is the integral of the speed :

$$\text{speed} = ds / st = || v || = \sqrt{v \cdot v}$$

$$(\text{arc}) \text{ length} = s(t) = \int_0^t || v(\tau) || \, d\tau$$

- assuming the curve starts at t = 0

In[35]:=
```
speed = Simplify[Sqrt[v . v]]
```

Out[35]= $\sqrt{1 + 4t^2 + 9t^4}$

$$\textbf{arclength} = \int_0^t \sqrt{1 + 4t^2 + 9t^4} \, d\tau$$

The unit tangent is the unit vector in the dirrection of the velocity

T = v/ || v || = v /speed = v /(ds/dt)

In[36]:=
```
UnitTangent = Simplify[v / speed]
```

Out[36]= $$\left\{\frac{1}{\sqrt{1 + 4t^2 + 9t^4}}, \frac{2t}{\sqrt{1 + 4t^2 + 9t^4}}, \frac{3t^2}{\sqrt{1 + 4t^2 + 9t^4}}\right\}$$

The unit normal N is the direction of dT/ds, and the curvature κ is its length.
So dT/ds = κ N.
N and κ are almost never computed this way, but rather are computed via the chain rule:

dT/ds = (dT/dt)/(ds/st) = (dT/dt)/speed.
Giving:

N = (dT/dt)/ || dT/dt ||
and
κ = || dT/dt || / speed
Alternatively,

κ = || v × a || / speed^3

(this comes from writing the acceleration in terms of the tangential and normal components)

In[37]:=
```
dTdt = Simplify[D[UnitTangent, t]]
```

Out[37]=
$$\left\{-\frac{2\,t\,(2+9\,t^2)}{(1+4\,t^2+9\,t^4)^{3/2}},\ \frac{2-18\,t^4}{(1+4\,t^2+9\,t^4)^{3/2}},\ \frac{6\,t\,(1+2\,t^2)}{(1+4\,t^2+9\,t^4)^{3/2}}\right\}$$

In[52]:=
```
? Assuming
```

In[53]:=
```
UnitNormal = Assuming[t ∈ Reals, Simplify[ dTdt / Sqrt[dTdt . dTdt] ]]
```

Out[53]=
$$\left\{-\frac{t\,(2+9\,t^2)}{\sqrt{(1+4\,t^2+9\,t^4)\,(1+9\,t^2+9\,t^4)}},\ \frac{1-9\,t^4}{\sqrt{(1+4\,t^2+9\,t^4)\,(1+9\,t^2+9\,t^4)}},\ \frac{3\,(t+2\,t^3)}{\sqrt{(1+4\,t^2+9\,t^4)\,(1+9\,t^2+9\,t^4)}}\right\}$$

In[54]:=
```
κ = Assuming[t ∈ Reals, Simplify[ Sqrt[dTdt . dTdt] / speed]]
```

Out[54]=
$$2\sqrt{\frac{1+9\,t^2+9\,t^4}{(1+4\,t^2+9\,t^4)^3}}$$

In[40]:=
```
VCrossA = Cross[v, a]
```

Out[40]=
$$\{6\,t^2,\ -6\,t,\ 2\}$$

In[41]:=
```
Simplify[Sqrt[VCrossA . VCrossA] / speed^3]
```

Out[41]=
$$\frac{2\sqrt{1+9\,t^2+9\,t^4}}{(1+4\,t^2+9\,t^4)^{3/2}}$$

The acceleration can be written in terms of the unit tangent and the unit normal, using the tangential component a_T and the normal component a_N:

a = a_TT + a_NN.

$$a_T = \frac{d^2 s}{dt^2}$$

$$a_N = \kappa\left(\frac{ds}{dt}\right)^2 = \sqrt{(\|a\|)^2 - a_T{}^2} = \|v\times a\|/\text{speed}$$

In[42]:=
```
aT = D[speed, t]
```

Out[42]=
$$\frac{8\,t+36\,t^3}{2\sqrt{1+4\,t^2+9\,t^4}}$$

In[43]:= `aN = Simplify[ Sqrt[(a . a) - aT^2]]`

Out[43]= $2\sqrt{\frac{1+9t^2+9t^4}{1+4t^2+9t^4}}$

In[44]:= `aN = Simplify[ Sqrt[VCrossA . VCrossA] / speed]`

Out[44]= $\frac{2\sqrt{1+9t^2+9t^4}}{\sqrt{1+4t^2+9t^4}}$

The Binormal is T* N, and the torsion is defined by
τ = - N ·dB/ds
= -N · (dB/dt)/speed.

There are other ways to compute the torsion, which can , for instance be found on mathworld.

In[55]:= `UnitBinormal = Assuming[t ∈ Reals, Simplify[Cross[UnitTangent, UnitNormal]]]`

Out[55]= $\left\{\frac{3t^2}{\sqrt{1+9t^2+9t^4}}, -\frac{3t}{\sqrt{1+9t^2+9t^4}}, \frac{1}{\sqrt{1+9t^2+9t^4}}\right\}$

In[50]:= `τ = - Simplify[UnitNormal . D[UnitBinormal, t] / speed]`

Out[50]= $\frac{3}{1+9t^2+9t^4}$